# Maintainer: Daniel M. Capella <polyzen@archlinux.org>
# Contributor: Florian Pritz <flo@xinu.at>
# Contributor: toliman <altergan@hs-albsig.de>

pkgname=tipp10
pkgver=3.3.1
pkgrel=1
pkgdesc='Intelligent touch typing tutor'
arch=('loong64' 'x86_64')
url=https://gitlab.com/tipp10/tipp10
license=('GPL-2.0-only')
depends=('qt6-multimedia')
makedepends=('cmake' 'git' 'qt6-tools')
source=("git+$url.git#tag=v$pkgver")
b2sums=('SKIP')

build() {
  cd tipp10
  cmake -DCMAKE_INSTALL_PREFIX=/usr .
  make
}

package() {
  cd tipp10
  make DESTDIR="$pkgdir" install
}
